;;; Mudsync --- Live hackable MUD
;;; Copyright © 2016 Christopher Allan Webber <cwebber@dustycloud.org>
;;;
;;; This file is part of Mudsync.
;;;
;;; Mudsync is free software; you can redistribute it and/or modify it
;;; under the terms of the GNU General Public License as published by
;;; the Free Software Foundation; either version 3 of the License, or
;;; (at your option) any later version.
;;;
;;; Mudsync is distributed in the hope that it will be useful, but
;;; WITHOUT ANY WARRANTY; without even the implied warranty of
;;; MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the GNU
;;; General Public License for more details.
;;;
;;; You should have received a copy of the GNU General Public License
;;; along with Mudsync.  If not, see <http://www.gnu.org/licenses/>.

;;; Game actor
;;; ==========

(define-module (mudsync gameobj)
  #:use-module (mudsync command)
  #:use-module (8sync actors)
  #:use-module (8sync agenda)
  #:use-module (srfi srfi-1)
  #:use-module (ice-9 format)
  #:use-module (ice-9 match)
  #:use-module (oop goops)
  #:export (<gameobj>

            gameobj-loc
            gameobj-gm

            gameobj-act-init
            gameobj-set-loc!
            gameobj-occupants
            gameobj-self-destruct

            slot-ref-maybe-runcheck
            val-or-run

            dyn-ref))

;;; Gameobj
;;; =======


;;; *all* game components that talk to players should somehow
;;; derive from this class.
;;; And all of them need a GM!

(define-class <gameobj> (<actor>)
  ;; location id
  (loc #:init-value #f
       #:getter gameobj-loc)
  
  ;; Uses a hash table like a set (values ignored)
  (occupants #:init-thunk make-hash-table)

  ;; game master id
  (gm #:init-keyword #:gm
      #:getter gameobj-gm)
  ;; a name to be known by
  (name #:init-keyword #:name
        #:init-value #f)
  (goes-by #:init-keyword #:goes-by
           #:init-value #f)

  (desc #:init-value #f
        #:init-keyword #:desc)

  ;; Commands we can handle
  (commands #:init-value '())

  ;; Commands we can handle by being something's container
  (container-commands #:init-value '())

  ;; Commands we can handle by being contained by something else
  (contained-commands #:init-value '())

  ;; Most objects are generally visible by default
  (generally-visible #:init-value #t
                     #:init-keyword #:generally-visible)
  ;; @@: Would be preferable to be using generic methods for this...
  ;;   Hopefully we can port this to Guile 2.2 soon...
  (visible-to-player?
   #:init-value (wrap-apply gameobj-visible-to-player?))

  ;; Set this on self-destruct
  ;; (checked by some "long running" game routines)
  (destructed #:init-value #f)

  (actions #:allocation #:each-subclass
           ;;; Actions supported by all gameobj
           #:init-value
           (build-actions
            (init gameobj-act-init)
            ;; Commands for co-occupants
            (get-commands gameobj-get-commands)
            ;; Commands for participants in a room
            (get-container-commands gameobj-get-container-commands)
            ;; Commands for inventory items, etc (occupants of the gameobj commanding)
            (get-contained-commands gameobj-get-contained-commands)
            (get-occupants gameobj-get-occupants)
            (add-occupant! gameobj-add-occupant!)
            (remove-occupant! gameobj-remove-occupant!)
            (get-loc gameobj-act-get-loc)
            (set-loc! gameobj-act-set-loc!)
            (get-name gameobj-get-name)
            (set-name! gameobj-act-set-name!)
            (get-desc gameobj-get-desc)
            (goes-by gameobj-act-goes-by)
            (visible-name gameobj-visible-name)
            (self-destruct gameobj-act-self-destruct)
            (tell gameobj-tell-no-op)
            (assist-replace gameobj-act-assist-replace))))


;;; gameobj message handlers
;;; ========================

;; Kind of a useful utility, maybe?
(define (simple-slot-getter slot)
  (lambda (actor message)
    (<-reply message (slot-ref actor slot))))

(define (gameobj-replace-step-occupants actor occupants)
  ;; Snarf all the occupants!
  (display "replacing occupant\n")
  (when occupants
    (for-each
     (lambda (occupant)
       (<-wait occupant 'set-loc!
               #:loc (actor-id actor)))
     occupants)))

(define gameobj-replace-steps*
  (list gameobj-replace-step-occupants))

(define (run-replacement actor replaces replace-steps)
  (when replaces
    (mbody-receive (_ #:key occupants)
        (<-wait replaces 'assist-replace)
      (for-each
       (lambda (replace-step)
         (replace-step actor occupants))
       replace-steps))))

;; @@: This could be kind of a messy way of doing gameobj-act-init
;;   stuff.  If only we had generic methods :(
(define* (gameobj-act-init actor message #:key replace)
  "Your most basic game object init procedure.
Assists in its replacement of occupants if necessary and nothing else."
  (run-replacement actor replace gameobj-replace-steps*))

(define (gameobj-goes-by gameobj)
  "Find the name we go by.  Defaults to #:name if nothing else provided."
  (cond ((slot-ref gameobj 'goes-by) =>
         identity)
        ((slot-ref gameobj 'name) =>
         (lambda (name)
           (list name)))
        (else '())))

(define (gameobj-act-goes-by actor message)
  "Reply to a message requesting what we go by."
  (<-reply message #:goes-by (gameobj-goes-by actor)))

(define (val-or-run val-or-proc)
  "Evaluate if a procedure, or just return otherwise"
  (if (procedure? val-or-proc)
      (val-or-proc)
      val-or-proc))

(define (filter-commands commands verb)
  (filter
   (lambda (cmd)
     (equal? (command-verbs cmd)
             verb))
   commands))

(define* (gameobj-get-commands actor message #:key verb)
  "Get commands a co-occupant of the room might execute for VERB"
  (define filtered-commands
    (filter-commands (val-or-run (slot-ref actor 'commands))
                     verb))
  (<-reply message
           #:commands filtered-commands
           #:goes-by (gameobj-goes-by actor)))

(define* (gameobj-get-container-commands actor message #:key verb)
  "Get commands as the container / room of message's sender"
  (define filtered-commands
    (filter-commands (val-or-run (slot-ref actor 'container-commands))
                     verb))
  (<-reply message #:commands filtered-commands))

(define* (gameobj-get-contained-commands actor message #:key verb)
  "Get commands as being contained (eg inventory) of commanding gameobj"
  (define filtered-commands
    (filter-commands (val-or-run (slot-ref actor 'contained-commands))
                     verb))
  (<-reply message
           #:commands filtered-commands
           #:goes-by (gameobj-goes-by actor)))

(define* (gameobj-add-occupant! actor message #:key who)
  "Add an actor to our list of present occupants"
  (hash-set! (slot-ref actor 'occupants)
             who #t))

(define* (gameobj-remove-occupant! actor message #:key who)
  "Remove an occupant from the room."
  (hash-remove! (slot-ref actor 'occupants) who))

(define* (gameobj-occupants gameobj #:key exclude)
  (hash-fold
   (lambda (occupant _ prev)
     (define exclude-it?
       (match exclude
         ;; Empty list and #f are non-exclusion
         (() #f)
         (#f #f)
         ;; A list of addresses... since our address object is (annoyingly)
         ;; currently a simple cons cell...
         ((exclude-1 ... exclude-rest)
          (member occupant exclude))
         ;; Must be an individual address!
         (_ (equal? occupant exclude))))
     (if exclude-it?
         prev
         (cons occupant prev)))
   '()
   (slot-ref gameobj 'occupants)))

(define* (gameobj-get-occupants actor message #:key exclude)
  "Get all present occupants of the room."
  (define occupants
    (gameobj-occupants actor #:exclude exclude))

  (<-reply message #:occupants occupants))

(define (gameobj-act-get-loc actor message)
  (<-reply message (slot-ref actor 'loc)))

(define (gameobj-set-loc! gameobj loc)
  "Set the location of this object."
  (define old-loc (gameobj-loc gameobj))
  (format #t "DEBUG: Location set to ~s for ~s\n"
          loc (actor-id-actor gameobj))

  (when (not (equal? old-loc loc))
    (slot-set! gameobj 'loc loc)
    ;; Change registation of where we currently are
    (if old-loc
        (<-wait old-loc 'remove-occupant! #:who (actor-id gameobj)))
    (if loc
        (<-wait loc 'add-occupant! #:who (actor-id gameobj)))))

;; @@: Should it really be #:id ?  Maybe #:loc-id or #:loc?
(define* (gameobj-act-set-loc! actor message #:key loc)
  "Action routine to set the location."
  (gameobj-set-loc! actor loc))

(define (slot-ref-maybe-runcheck gameobj slot whos-asking)
  "Do a slot-ref on gameobj, evaluating it including ourselves
and whos-asking, and see if we should just return it or run it."
  (match (slot-ref gameobj slot)
    ((? procedure? slot-val-proc)
     (slot-val-proc gameobj whos-asking))
    (anything-else anything-else)))

(define gameobj-get-name (simple-slot-getter 'name))

(define* (gameobj-act-set-name! actor message val)
  (slot-set! actor 'name val))

(define* (gameobj-get-desc actor message #:key whos-looking)
  (define desc-text
    (match (slot-ref actor 'desc)
      ((? procedure? desc-proc)
       (desc-proc actor whos-looking))
      (desc desc)))
  (<-reply message desc-text))

(define (gameobj-visible-to-player? gameobj whos-looking)
  "Check to see whether we're visible to the player or not.
By default, this is whether or not the generally-visible flag is set."
  (slot-ref gameobj 'generally-visible))

(define* (gameobj-visible-name actor message #:key whos-looking)
  ;; Are we visible?
  (define we-are-visible
    ((slot-ref actor 'visible-to-player?) actor whos-looking))

  (define name-to-return
    (if we-are-visible
        ;; Return our name
        (match (slot-ref actor 'name)
          ((? procedure? name-proc)
           (name-proc actor whos-looking))
          ((? string? name)
           name)
          (#f #f))
        #f))
  (<-reply message #:text name-to-return))

(define (gameobj-self-destruct gameobj)
  "General gameobj self destruction routine"
  ;; Unregister from being in any particular room
  (gameobj-set-loc! gameobj #f)
  (slot-set! gameobj 'destructed #t)
  ;; Boom!
  (self-destruct gameobj))

(define (gameobj-act-self-destruct gameobj message)
  "Action routine for self destruction"
  (gameobj-self-destruct gameobj))

;; Unless an actor has a tell message, we just ignore it
(define gameobj-tell-no-op
  (const 'no-op))

(define (gameobj-replace-data-occupants actor)
  "The general purpose list of replacement data"
  (list #:occupants (hash-map->list (lambda (occupant _) occupant)
                                    (slot-ref actor 'occupants))))

(define (gameobj-replace-data* actor)
  ;; For now, just call gameobj-replace-data-occupants.
  ;; But there may be more in the future!
  (gameobj-replace-data-occupants actor))

;; So sad that objects must assist in their replacement ;_;
;; But that's life in a live hacked game!
(define (gameobj-act-assist-replace actor message)
  "Vanilla method for assisting in self-replacement for live hacking"
  (apply <-reply message
         (gameobj-replace-data* actor)))


;;; Utilities every gameobj has
;;; ---------------------------

(define (dyn-ref gameobj special-symbol)
  "Dynamically look up a special object from the gm"
  (match special-symbol
    ;; if it's a symbol, look it up dynamically
    ((? symbol? _)
     (mbody-val (<-wait (slot-ref gameobj 'gm) 'lookup-special
                        #:symbol special-symbol)))
    ;; if it's false, return nothing
    (#f #f)
    ;; otherwise it's probably an address, return it as-is
    (_ special-symbol)))
